Может кто нибудь подскажет, вопрос такой:
В Справочнике ВнутренниеДокументы добавлены несколько дополнительных реквизитов с разными типами. Некоторые заполнены некоторые нет.
Цель - получить все эти реквизиты в запросе.

Запрос следующий:

```1c
    ВЫБРАТЬ
    | ДополнительныеРеквизитыИСведения.Ссылка КАК Ссылка
    |ПОМЕСТИТЬ ВТ_Свойства
    |ИЗ
    | ПланВидовХарактеристик.ДополнительныеРеквизитыИСведения КАК ДополнительныеРеквизитыИСведения
    |ГДЕ
    | ДополнительныеРеквизитыИСведения.НаборСвойств = &НаборСвойств
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ВнутренниеДокументыДополнительныеРеквизиты.Свойство КАК Свойство,
    | ВнутренниеДокументыДополнительныеРеквизиты.Значение КАК Значение
    |ИЗ
    | ВТ_Свойства КАК ВТ_Свойства
    | ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ВнутренниеДокументы.ДополнительныеРеквизиты КАК ВнутренниеДокументыДополнительныеРеквизиты
    | ПО ВТ_Свойства.Ссылка = ВнутренниеДокументыДополнительныеРеквизиты.Свойство
```

При этом в результате запроса выводятся только заполненные реквизиты. Пробовал разные соединения, проверку на null, ничего не помогает.
---
Так Вы и ничего не сделали из того, что Вам порекомендовали. Еще раз: в последнем запросе пакета свойство нужно брать не из временной таблицы документа, а из временной таблицы плана видов характеристик.
---
Полагаю, здесь ошибка - это выборка поля "Свойство" из ТЧ документа, а не из временной таблицы.
---
Плюс не учитываются общие дополнительные реквизиты и сведения, у которых свойство "НаборСвойств" не заполняется. Чтобы их учесть, если не ошибаюсь, корректнее написать первый запрос пакета так:

```1c
    ВЫБРАТЬ
    ДопРеквизиты.Свойство
    ИЗ
    Справочник.НаборыДополнительныхРеквизитовИСведений.ДополнительныеРеквизиты КАК ДопРеквизиты
    ГДЕ
    ДопРеквизиты.Ссылка = &НаборСвойств

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    ДопСведения.Свойство
    ИЗ
    Справочник.НаборыДополнительныхРеквизитовИСведений.ДополнительныеСведения КАК ДопСведения
    ГДЕ
    ДопСведения.Ссылка = &НаборСвойств
```

Ну а если есть возможность, используйте лучше программный интерфейс подсистемы (общий модуль "УправлениеСвойствами").
---
ну да скорее всего если обращаться к временной таблице все получится
---
```1c
    ВЫБРАТЬ
    | ДополнительныеРеквизитыИСведения.Ссылка КАК Ссылка
    |ПОМЕСТИТЬ ВТ_Свойства
    |ИЗ
    | ПланВидовХарактеристик.ДополнительныеРеквизитыИСведения КАК ДополнительныеРеквизитыИСведения
    |ГДЕ
    | ДополнительныеРеквизитыИСведения.НаборСвойств = &НаборСвойств
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ВнутренниеДокументыДополнительныеРеквизиты.Свойство КАК Свойство,
    | ВнутренниеДокументыДополнительныеРеквизиты.Значение КАК Значение
    |ПОМЕСТИТЬ ВТ_СвойстваДок
    |ИЗ
    | Справочник.ВнутренниеДокументы.ДополнительныеРеквизиты КАК ВнутренниеДокументыДополнительныеРеквизиты
    |ГДЕ
    | ВнутренниеДокументыДополнительныеРеквизиты.Ссылка = &Док
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | ВТ_СвойстваДок.Свойство КАК Свойство,
    | ВТ_СвойстваДок.Значение КАК Значение
    |ИЗ
    | ВТ_Свойства КАК ВТ_Свойства
    | ЛЕВОЕ СОЕДИНЕНИЕ ВТ_СвойстваДок КАК ВТ_СвойстваДок
    | ПО ВТ_Свойства.Ссылка = ВТ_СвойстваДок.Свойство
```

Теперь снизу появилось два пустых свойства, а те которые мне нужны (тест12111 и епрекрр) так и не вышли с пустыми значениями
---
Так Вы и ничего не сделали из того, что Вам порекомендовали. Еще раз: в последнем запросе пакета свойство нужно брать не из временной таблицы документа, а из временной таблицы плана видов характеристик.
---
Спасибо огромное, все получилось!)
---
в таблицу ВТ_СвойстваДок попадают почему то только заполненные реквизиты
---
а как теперь связать эти доп реквизиты с документом?
---
Как-то так:

```1c
    ВЫБРАТЬ
    ВнутренниеДокументы.Ссылка КАК Документ,
    ВТ_Свойства.Ссылка КАК Свойство,
    ДопРеквизиты.Значение КАК Значение
    ИЗ
    Справочник.ВнутренниеДокументы КАК ВнутренниеДокументы
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ВТ_Свойства КАК ВТ_Свойства
    ПО ИСТИНА
    ЛЕВОЕ СОЕДИЕНИ Справочник.ВнутренниеДокументы.ДополнительныеРеквизиты КАК ДопРеквизиты
    ПО ВнутренниеДокументы.Ссылка = ДопРеквизиты.Ссылка
    И ВТ_Свойства.Ссылка = ДопРеквизиты.Свойство
    ГДЕ
    ВнутренниеДокументы.Ссылка = &Док
```
---
Можно все одним запросом:

```1c
    ВЫБРАТЬ
    ВнутренниеДокументы.Ссылка КАК Документ,
    ТаблицаСвойств.Ссылка КАК Свойство,
    ТаблицаЗначений.Значение КАК Значение
    ИЗ
    Справочник.ВнутренниеДокументы КАК ВнутренниеДокументы
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ПланВидовХарактеристик.ДополнительныеРеквизитыИСведения КАК ТаблицаСвойств
    ПО ТаблицаСвойств.НаборСвойств = &НаборСвойств
    ЛЕВОЕ СОЕДИЕНИЕ Справочник.ВнутренниеДокументы.ДополнительныеРеквизиты КАК ТаблицаЗначений
    ПО ВнутренниеДокументы.Ссылка = ТаблицаЗначений.Ссылка
    И ТаблицаСвойств.Ссылка = ТаблицаЗначений.Свойство
    ГДЕ
    ВнутренниеДокументы.Ссылка = &Док
```
---
Просто спаситель!!! Очень вовремя!!! Спасибо огромное вам!
---
большое спасибо! В прошлый раз выкрутилась как-то с помощью вложенных запросов, теперь сделала всё красиво))
---
Опечатка у вас в тексте " ЛЕВОЕ СОЕДИЕНИЕ". Правильно "ЛЕВОЕ СОЕДИНЕНИЕ". А так все ок. спасибо)
---
НаборСвойств что нужно передавать в переменную "Набор свойств"?
---
Еще раз Спасибо! Очень помогли